MS 365 Excel Basics 05:

IF Function and Logical Test. IFS, IFNA, OR, AND, ISNUMBER Functions & More!

Table of Contents

IF Function arguments



Logical Tests

- A **logical test** is an expression (formula) that evaluates to one of only two possible values TRUE or FALSE.
- TRUE and FALSE values are called **logical values** or **Boolean values** (after mathematician George Boole).
- A logical test can have one or more requirements called **conditions** or **criteria**.
    - If you ask the question: "Are sales greater than or equal to 50,000?", the condition is **>=50000**.
    - If you ask the question "Is the value a number?", the condition is **Is the data type a number?**
- You can use comparative operators (>, >=, <, <=, =, <>) to create individual logical tests.
    - When you use comparative operators, you place the operator directly between values such as: 65000>50000 or G20>G19. This is different than with the SUMIFS and similar functions, where the comparative operator is a text value. Example:

	B	C	D	E	F	G	H	I	J	K
2	**Example 1:**				**COUNTIFS, SUMIFS and the like use "text" Comparative Operators:**					
3	Are sales greater than or equal to the bonus hurdle?				How many sales are 50,000 or more?					
4										
5	Logical Test:	Sales >= Hurdle?			Sales		Condition:	>=50000		
6	Condition:	>=50000			55000		Count:	2		
7					25000					
8	Your Sales ($)	65,000			68500		Text value used in COUNTIFS. Direct logical tests cannot use text conditions.			
9	Bonus Hurdle ($)	50,000			43050					
10	65,000>50,000?	TRUE	C10: =C8>=C9				I6: =COUNTIFS(F6:F9,I5)			

    - You can use IS functions (like ISNUMBER or ISTEXT) to create individual logical tests. Example:

	B	C	D	E
12	**Example 2:**			
13	Why is total incorrect?			
15	Logical Test:	Is Sales Value A Number?		
16	Condition:	Is the Data Type a Number?		
18	Sales ($)	ISNUMBER		
19	250	TRUE	=ISNUMBER(B19:B21)	
20	500	FALSE		
21	100	TRUE		
22	350			

- You can use the aggregate functions AND, OR or NOT to create AND, OR or NOT Logical Tests. Example:

Example 3:

To earn credit, the customer must have last year sales of 1,000,000 or more AND have a credit rating of more than 4.

Logical Test: Are Sales Last Year >=1000000 AND Credit Rating > 4?

Conditions: Condition 1: >=1000000

Condition 2: >4

Sales Last Year	1000000
Credit Rating	4

Customer	Sales Last Year	Rating	Pass Both Tests?
Safeway	2000000	3.5	FALSE
QFC	1500000	5.5	TRUE

E36: =AND(C36>=$D$32,D36>$D$33)

- Except in the Power Query tool, logical tests are not case-sensitive, so "Quad" = "quad" = TRUE. Example:

Example 4:

Is the product a "Quad"?

Condition	Quad

Logical Test: Product = "Quad"

Condition: =Quad

Product	Quad?
quad	TRUE
Aspen	FALSE
Quad	TRUE

C48: =B48:B50=C42

- Except in the Power Query tool, any non-zero number is interpreted as TRUE and zero is interpreted as FALSE.
- Any math operation on logical values converts TRUE to 1 and FALSE to 0 (zero). Example:

Example 5:

Are both numbers non-zeros? What happens if you add zero to the logical values?

Logical Test: Number1<>0 AND Number2<>0

Condition 1: Number1<>0

Number2<>0

Number 1	Number 2	AND	AND	Add 0
2	0	FALSE	FALSE	0
0	0	FALSE	FALSE	0
-1	43	TRUE	TRUE	1

Logical Test with NOT operator: D60: =AND(B60<>0,C60<>0)

Logical Test that uses numbers as TRUE or FALSE: E60: =AND(B60:C60)

Add 0 to the logical values to convert to 1 and 0: F60: =E60:E62+0

Types of Logical Tests

- Single Condition Logical Test = Single condition must match.
- NOT Logical Test = Checks whether two items are not equal. A NOT Logical Test will also convert a TRUE to FALSE and a FALSE to TRUE.
- OR Logical Test = Run two or more logical tests and one or more tests must equal TRUE for the OR Logical Test to deliver a TRUE.
    - Four possibilities for an OR Logical Test with two tests:
        - TRUE, TRUE = TRUE
        - FALSE, TRUE = TRUE
        - TRUE, FALSE = TRUE
        - FALSE, FALSE = FALSE
    - The math operator for an OR Logical Test is the plus operator: +.
        - TRUE + TRUE = 1 + 1 = 2
        - FALSE + TRUE = 0 + 1 = 1
        - TRUE + FALSE = 1 + 0 = 1
        - FALSE + FALSE = 0 + 0 = 0
- AND Logical Test = Run two or more logical tests and all tests must equal TRUE for the AND Logical Test to deliver a TRUE.
    - Four possibilities for an AND Logical Test with two tests:
        - TRUE, TRUE = TRUE
        - TRUE, FALSE = FALSE
        - FALSE, TRUE = FALSE
        - FALSE, FALSE = FALSE
    - The math operator for an AND Logical Test is the multiplication operator: *
        - TRUE * TRUE = 1 * 1 = 1
        - TRUE * FALSE = 1 * 0 = 0
        - FALSE * TRUE = 0 * 1 = 0
        - FALSE * FALSE = 0 * 0 = 0
- BETWEEN Logical Test = Is a type of AND Logical Test that tests whether a value is between a lower and upper limit ,like: *Is 15 between 10 and 19?*

Comparative Operators

Comparative Operator:	=	>	>=	<	<=	<>
Possible Words:	equal	greater than	greater than or equal to	less than	less than or equal to	not
		more than	at least	below	at most	complement of
		above	no less than	under	no more than	
			X or more		X or less	
Examples of Words:	equals 2000	greater than 2000	greater than or equal to 2000	less than 2000	less than or equal to 2000	not 2000
		more than 2000	at least 2000	below 2000	at most 2000	complement of 2000
		above 2000	no less than 2000	under 2000	no more than 2000	
			2000 or more		2000 or less	

Hurdle:
2000

IS Functions

IS function	Logical question that function asks
ISNUMBER	Is value a number?
ISTEXT	Is value text?
ISBLANK	Is cell empty cell?
ISNA	Is value an #N/A error?
ISERR	Is value an error, except for #N/A?
ISERROR	Is value an error? (#DIV/0!, #REF!, #NAME?, #N/A, #VALUE!, #NULL!, #NUM!, #SPILL!, #CALC!, #BUSY!).
ISNONTEXT	Is value NOT text?
ISLOGICAL	Is value a logical value?
ISFORMULA	Does cell contains a formula?
ISREF	Is the value a reference?
ISEVEN	Is number even?
ISODD	Is number odd?

AND, OR and NOT Function arguments







AND and OR Logical Test example from video #2

Single condition logical test: Mom says: "If you take out the garbage, you get dessert".
If "take out garbage" = TRUE, you get dessert

Two condition logical test: Mom says: "If you take out the garbage AND clean the table, you get dessert".
If "take out garbage" = TRUE AND "clean the table" = TRUE, you get dessert
TRUE, TRUE = you get dessert

AND Logical Test =

Two or more logical tests are used to test whether to count or add an item.
All tests must be met, for the item to be included.

Four possibilities for an AND Logical Test with two tests:

TRUE, TRUE = TRUE = you get dessert
TRUE, FALSE = FALSE = No dessert
FALSE, TRUE = FALSE = No dessert
FALSE, FALSE = FALSE = No dessert

Two condition logical test: Mom says: "If you take out the garbage OR clean the table, you get dessert".
You get dessert if you just take out garbage, or you just clean the table, or you do both!

OR Logical Test

One or more tests needs to come out true in order to count or add an item.
You must get at leaset one TRUE, for the item to be included.

Four possibilities for an OR Logical Test with two tests:

TRUE, TRUE = TRUE 2 TRUEs
FALSE, TRUE = TRUE 1 TRUE
TRUE, FALSE = TRUE 1 TRUE
FALSE, FALSE = FALSE 0 TRUE

IF Function to deliver one of two items

IF to Deliver 1 of 2 Items

Ex 6 Contract reads: If employee has sales of $50,000 or more they get a bonus.

1) Did employee get bonus? TRUE or FALSE?

2) Are your sales greater than or equal to $50,000? TRUE gets you $500.00, FALSE gets you $0

Bonus Amount ($)	500
Bonus Hurdle ($)	50,000
Your Sales ($)	65,000
Do You Get Bonus?	TRUE
Your Bonus ($)	500

C11: =C10>=C9

C12: =IF(C10>=C9,C8,0)

* IF delivers one of two numbers to the cell

Ex 7 1) Do Debits = Credits? TRUE or FALSE?

2) If Debits = Credits, show the text "In Balance", if not "NOT In Balance"

Debit Numbers	Credit Numbers
$58	$58
$43	$43
$21	$21
$122	$122

In Balance?	FALSE
Message?	NOT In Balance

C25: =B23=C23

C26: =IF(B23=C23,"In Balance","NOT In Balance")

* IF delivers one of two text values to the cell

Ex 8 Contract Reads if you have sales of more than $30,000, you earn a 5% bonus, otherwise you get a 1% bonus.

1) Create a spilled formula that will calculate the paid bonus.

Hurdle	Bonus %	No Bonus %
30,000.00	0.05	0.01

Sales	Calculate Commission Paid
17,382.00	173.82
19,504.00	195.04
29,999.99	300.00
30,000.01	1,500.00
16,081.00	160.81

C36: =IF(B36:B40>B33,C33,D33)*B36:B40

* IF delivers one or two numbers to the formula

Use IF Function in an Invoice and use “” A Zero Length Text String

Ex 9 **Task:** Create Invoice that can lookup price and calculate sales based on whether data is entered into a cell.

Goal: Use IF to deliver 1 of 2 items, either a formula or "" (show nothing).

Product	Units Sold	Price ($)	Sales ($)
Yanaki	12	27.95	335.4
Sunshine	44	19.95	877.8
Aspen	34	25.95	882.3
Carlota	12	26.95	323.4

Product	Price
Quad	43.95
Carlota	26.95
Aspen	25.95
Yanaki	27.95
Sunshine	19.95
FastCatch	31.95

Prove that "" is a Zero Length Text String:

Empty cell =>		TRUE	Formula in cell D17: =ISBLANK(C17)
"" =>		FALSE	Formula in cell D18: =ISBLANK(C18)
"" =>		TRUE	Formula in cell D19: =ISTEXT(C19)
"" =>		0	Formula in cell D20: =LEN(C20)

IF with AND function

Ex 10 Task: Students are eligible for scholarship if:
They have completed 45 or more credits AND have GPA more than 2.5.
Goal: Create formula that shows "Eligible" or "Not Eligible"

Credit Hurdle:	45
GPA Hurdle:	2.5

Yes:	Eligible
No:	Not Eligible

Student	Start Date	Major	Credits	GPA	Eligible?	Eligible?
Carey, Zada	9/29/2020	Business	45	1.7	Not Eligible	Not Eligible
Emmons, Christi	7/14/2018	Accounting	135	2.3	Not Eligible	Not Eligible
Lear, Vania	9/3/2020	Chemistry	45	3	Eligible	Eligible
Meador, Corazon	11/21/2019	Accounting	90	3.1	Eligible	Eligible
Mohamed, Abdi	1/28/2021	Business	23	1.6	Not Eligible	Not Eligible
Nga, Luong	7/7/2020	Physics	45	2.4	Not Eligible	Not Eligible
Robinson, Chantel	4/12/2020	History	70	4	Eligible	Eligible
Rouse, Sioux	6/30/2020	Chemistry	40	2.4	Not Eligible	Not Eligible
Simone, Alanna	8/2/2019	Physics	60	3.5	Eligible	Eligible
Thornburg, Tyrone	12/27/2019	Sociology	75	3.9	Eligible	Eligible

G10: =IF(AND(E10>=$C$6,F10>$C$7),$H$6,$H$7)
H10: =IF((E10:E19>=C6)*(F10:F19>C7),H6,H7)

IF with OR Function

Ex 11 Task: A company only extends credit to customers who have:
Last Year Sales > 1,000,000 OR Credit Rating >=4.
Goal 1: Create formula to show "Credit" in each row where customer passes OR Logical Test.
Goal 2: Create a formula that shows "None" in each row if the customer does not pass either rule.

Last Year Sales >	1,000,000
Credit Rating >=	4

Last Year Sales	Credit Rating	Customer	Credit?	Credit?	Customers that Pass Neither Rule	Customers that Pass Neither Rule
1,250,000	4.6	SW	Credit	Credit		
955,500	3.7	PCC			None	None
875,000	4	QFC	Credit	Credit		
2,100,500	2	FM	Credit	Credit		
550,750	1.6	WM			None	None
2,500,000	2.9	L	Credit	Credit		

E11: =IF(OR(B11>$C$7,C11>=$C$8),"Credit","")
F11: =IF((B11:B16>C7)+(C11:C16>=C8),"Credit","")
G11: =IF(NOT(OR(B11>$C$7,C11>=$C$8)),"None","")
H11: =IF(NOT((B11:B16>C7)+(C11:C16>=C8)),"None","")

IFS function

Ex 12 **Goal:** Report must have the label: "Net Loss", Net Income" or "Break Even", based on whether revenues are less than, equal to or bigger than expenses.
IFS to deliver 1 of 3 **Text Items** to the cell: "Net Loss" or "Break Even" or "Net Income"

Revenue	$65,000
COGS Expense	29,500
Operations Expense	12,750
Admin Expense	5,750
Other Expense	6,950
Total Expenses	$54,950
Net Income	$10,050

Formula in cell C11: =SUM(C7:C10)
Formula in cell C12: =ABS(C6-C11)
Formula in cell B12: =IFS(C6>C11,"Net Income",C11>C6,"Net Loss",TRUE,"Break Even")

```
=IFS(C6>C11,"Net Income",C11>C6,"Net Loss",TRUE,"Break Even")
```

IFS(logical_test1, value_if_true1, [logical_test2, value_if_true2], [logical_test3, **value_if_true3**], [logical_test4, ...)

Old School used 2 IF functions nested together like:

Net Income

B21: =IF(C6>C11,"Net Income",IF(C6<C11,"Net Loss","Break Even"))

IFNA function arguments

value argument = a value that might deleiver an #N/A error.

value_if_na argument = what to display when an #N/A error occurs

IFNA(value, value_if_na)

Goal: show value or an alternative when an #N/A error occurs.

IFNA function example

Ex 8 **Goal:** Show formula in cell if user creates a formula in that cell, otherwise show nothing.
Use IF to deliver 1 of 2 items, either a **formula** or **"" (show nothing)**.

Product	Price	Units	Units
Quad	43.95	32	$1,406.40
Carlota	26.95	25	$673.75
Aspen	25.95	44	$1,141.80

Formula in cell E6:

```
=C6:C8*D6:D8
```

Formula in cell G6:

```
=IFNA(FORMULATEXT(E6),"")
```

IFNA(**value**, value_if_na)

IFERROR Function arguments



When Not To Use the IFERROR function

Ex 9 **Goal:** Create array formula that calculates when 3 or more scores have been entered.

Formula in cell H17:

=SUM(LARGE(C17:G17,{1,2,3}))

Formula in cell I17:

=IFERROR(SUM(LARGE(C17:G17,{1,2,3})),"")

Formula in cell J17:

=IF(COUNT(C17:G17)>2,SUM(LARGE(C17:G17,{1,2,3})),"")

** Advantage is that the array formula does not have to be evaluated in every row: only rows where the logical test evaluates to TRUE. For large data sets, this saves calculation time and improves performance.

Thrower	Time 1	Time 2	Time 3	Time 4	Time 5	Add Top 3	Add Top 3	Add Top 3
Bower	0.00	86.87	92.40	80.89	85.09	264.36	264.36	264.36
Noline	72.09	108.27	73.70	93.34	0.00	275.31	275.31	275.31
Washington	55.26	85.96	89.82	34.85	31.66	231.04	231.04	231.04
						#NUM!		
						#NUM!		
						#NUM!		

Array Constant

Array Constant = Hard code an array of values into formula.

Array Syntax:

{ } House the array

, = Column

; = Row

Example {1,2,3} in the formula SUM(LARGE(C17:G17,{1,2,3} to fore the LARGE function to deliver the three largest values to the SUM function, so that SUM can add the three largest value.